

**HFLPDF****PURPOSE**

Compute the half-logistic or the generalized half-logistic probability density function.

**DESCRIPTION**

The half-logistic distribution has the following probability density function:

$$f(x) = \frac{2e^{-x}}{(1 + e^{-x})^2} \quad x \geq 0 \quad \text{(EQ Aux-201)}$$

The generalized form of this distribution has the probability density function:

$$f(x, k) = \frac{2\left(1 - kx^{\frac{1}{k}-1}\right)}{(1 + (1 - kx^{1/k}))^2} \quad 0 \leq x \leq \frac{1}{k}, k > 0 \quad \text{(EQ Aux-202)}$$

The half-logistic distribution is formed by folding the standard logistic distribution (that is, the distribution of ABS(x) where x has a logistic distribution).

**SYNTAX**

LET <y> = HFLPDF(<x>,<k>) <SUBSET/EXCEPT/FOR qualification>

where <x> is a number, parameter, or variable;

<k> is an optional number, parameter, or variable that specifies the shape parameter;

<y> is a variable or a parameter (depending on what <x> is) where the computed half-logistic pdf value is stored; and where the <SUBSET/EXCEPT/FOR qualification> is optional.

If the <k> parameter is omitted, the half-logistic pdf is computed. If <k> is given, the generalized half-logistic pdf is computed.

**EXAMPLES**

LET A = HFLPDF(3)

LET A = HFLPDF(0.8,0.4)

LET X2 = HFLPDF(X1)

LET X2 = HFLPDF(X1,K)

**NOTE**

DATAPLOT limits the value of the shape parameter to values less than or equal to 10.

**DEFAULT**

None

**SYNONYMS**

None

**RELATED COMMANDS**

HFLCDF	=	Compute the generalized half-logistic cumulative distribution function.
HFLPPF	=	Compute the generalized half-logistic percent point function.
LOGCDF	=	Compute the logistic cumulative distribution function.
LOGPDF	=	Compute the logistic probability density function.
LOGPPF	=	Compute the logistic percent point function.
LLGCDF	=	Compute the log-logistic cumulative distribution function.
LLGPDF	=	Compute the log-logistic probability density function.
LLGPPF	=	Compute the log-logistic percent point function.

**REFERENCE**

"Continuous Univariate Distributions - Vol. 2," 2nd. Ed., Johnson, Kotz, and Balakrishnan, John Wiley and Sons, 1994 (pp. 150-151).

**APPLICATIONS**

Lifetime Analysis

## IMPLEMENTATION DATE

95/10

## PROGRAM

```

MULTIPLY 2 2; MULTIPLY CORNER COORDINATES 0 0 100 100
TITLE AUTOMATIC
X1LABEL HALF LOGISTIC DISTRIBUTION
PLOT HFLPDF(X) FOR X = 0 0.01 5
LET K = 0.5
X1LABEL HALF-LOGISTIC DISTRIBUTION - K = ^K
LET UL = 1/K
PLOT HFLPDF(X,K) FOR X = 0 0.01 UL
LET K = 1
X1LABEL HALF-LOGISTIC DISTRIBUTION - K = ^K
LET UL = 1/K
PLOT HFLPDF(X,K) FOR X = 0 0.01 UL
LET K = 2
X1LABEL HALF-LOGISTIC DISTRIBUTION - K = ^K
LET UL = 1/K - 0.001
PLOT HFLPDF(X,K) FOR X = 0 0.001 UL
END OF MULTIPLY

```

